```
      SUBROUTINE LRPROP(D)
C        COMPUTES AREF, THE REFERENCE VALUE OF RADIO ATTENUATION
C          VERSION 1.2.1 (AUG 71/MAR 77/APR 79)
C             OF THE LONGLEY-RICE (1968) MODEL
C          PRINCIPAL CHANGES-
C             1.1. A SIMPLIFICATION OF THE LINE-OF-SIGHT AND SCATTER
C                  ROUTINES
C             1.2. A CHANGE IN THE LINE-OF-SIGHT ROUTINE AND IN THE
C                  SUBSEQUENT CALCULATIONS.  RESULTS ARE IMPROVED WHEN
C                  ONE OR BOTH ANTENNAS ARE HIGH.
C          VALID ONLY FOR...
C               FREQUENCIES BETWEEN 20 MHZ AND 20 GHZ
C               ANTENNA HEIGHTS BETWEEN 0.5 M AND 3000 M
C               ELEVATION ANGLES LESS THAN 200 MRAD
C
      COMMON/PROP/KWX,AREF,MDP,DIST,HG(2),WN,DH,ENS,GME,ZGND,
     X   HE(2),DL(2),THE(2)
         COMPLEX ZGND
      COMMON/PROPA/DLSA,DX,AEL,AK1,AK2,AED,EMD,AES,EMS,DLS(2),DLA,THA
C
      COMMON/SAVE/SAVA(6),WLOS,WSCAT,DMIN,XAE,SAVB(40)
C
      LOGICAL WLOS,WSCAT
C
      DATA THIRD/0.3333333/
C
       IF(MDP) 10,32,10
C
   10  CONTINUE
       DO 11 J=1,2
   11  DLS(J)=SQRT(2.*HE(J)/GME)
       DLSA=DLS(1)+DLS(2)
       DLA=DL(1)+DL(2)
       THA=AMAX1(THE(1)+THE(2),-DLA*GME)
       WLOS=.FALSE.
       WSCAT=.FALSE.
C
C        CHECK PARAMETER RANGES
          IF(ENS .LT. 250. .OR. ENS .GT. 400.) GO TO 154
          IF(GME .LT. 75E-9 .OR. GME .GT. 250E-9) GO TO 154
          IF(REAL(ZGND) .LE. ABS(AIMAG(ZGND))) GO TO 154
          DO 121 J=1,2
            IF(ABS(THE(J)) .GT. 200E-3) GO TO 153
            IF(DL(J) .LT. 0.1*DLS(J) .OR. DL(J) .GT. 3.*DLS(J))
     X         GO TO 153
  121     CONTINUE
          IF(WN .LT. 0.838 .OR. WN .GT. 210.) GO TO 151
          DO 122 J=1,2
            IF(HG(J) .LT. 1. .OR. HG(J) .GT. 1000.) GO TO 151
  122     CONTINUE
         GO TO 158
  153     KWX=MAX0(KWX,3)
  151     KWX=MAX0(KWX,1)
```

```
            IF(WN .LT. 0.419 .OR. WN .GT. 420.) GO TO 154
            DO 132 J=1,2
              IF(HG(J) .LT. 0.5 .OR. HG(J) .GT. 3000.) GO TO 154
  132       CONTINUE
       GO TO 158
  154    KWX=4
  158  CONTINUE
       DMIN=ABS(HE(1)-HE(2))/200E-3
C
C        COEFFICIENTS FOR THE DIFFRACTION RANGE
C
       Q=ADIFF(0.)
       XAE=(WN*GME**2)**(-THIRD)
       D3=AMAX1(DLSA,1.3787*XAE+DLA)
       D4=D3+2.7574*XAE
       A3=ADIFF(D3)
       A4=ADIFF(D4)
       EMD=(A4-A3)/(D4-D3)
       AED=A3-EMD*D3
C
       IF(MDP) 33,32,31
  31     MDP=0
  32     DIST=D
       IF(DIST .LE. 0.) GO TO 38
  33     CONTINUE
         IF(DIST .GT. 1000E3) KWX=MAX0(KWX,1)
         IF(DIST .LT. DMIN) KWX=MAX0(KWX,3)
         IF(DIST .LT. 1E3 .OR. DIST .GT. 2000E3) KWX=4
  38   CONTINUE
C
       IF(DIST .GE. DLSA) GO TO 50
C
       IF(WLOS) GO TO 48
C
C        COEFFICIENTS FOR THE LINE-OF-SIGHT RANGE
C
       Q=ALOS(0.)
       D2=DLSA
       A2=AED+D2*EMD
       D0=1.908*WN*HE(1)*HE(2)
       IF(AED .LT. 0.) GO TO 41
         D0=AMIN1(D0,0.5*DLA)
         D1=D0+0.25*(DLA-D0)
       GO TO 42
  41     D1=AMAX1(-AED/EMD,0.25*DLA)
  42   A1=ALOS(D1)
       IF(D0 .GE. D1) GO TO 43
         A0=ALOS(D0)
         Q=ALOG(D2/D0)
         AK2=AMAX1(0.,((D2-D0)*(A1-A0)-(D1-D0)*(A2-A0))/
     X        ((D2-D0)*ALOG(D1/D0)-(D1-D0)*Q))
       IF(AK2 .GT. 0.) GO TO 44
       IF(AED .GE. 0.) GO TO 44
```

```
 43       AK2=0.
          AK1=(A2-A1)/(D2-D1)
         IF(AK1 .GT. 0.) GO TO 46
         GO TO 45
 44       AK1=(A2-A0-AK2*Q)/(D2-D0)
          IF(AK1 .GE. 0.) GO TO 46
          AK1=0.
          AK2=DIM(A2,A0)/Q
         IF(AK2 .GT. 0.) GO TO 46
 45       AK1=EMD
 46      AEL=A2-AK1*D2-AK2*ALOG(D2)
         WLOS=.TRUE.
 48      IF(DIST .LE. 0.) GO TO 50
         AREF=AEL+AK1*DIST+AK2*ALOG(DIST)
         GO TO 60
C
 50      IF(WSCAT) GO TO 58
C
C          COEFFICIENTS FOR THE SCATTER RANGE
C
         Q=ASCAT(0.)
         D5=DLA+200E3
         D6=D5+200E3
         A6=ASCAT(D6)
         A5=ASCAT(D5)
         IF(A5 .LT. 1000.) GO TO 51
            EMS=EMD
            AES=AED
            DX=10E6
          GO TO 52
 51         EMS=(A6-A5)/200E3
            DX=AMAX1(DLSA,DLA+0.3*XAE*ALOG(47.7*WN),
     X          (A5-AED-EMS*D5)/(EMD-EMS))
            AES=(EMD-EMS)*DX+AED
 52      WSCAT=.TRUE.
C
 58      IF(DIST .GT. DX) GO TO 59
            AREF=AED+EMD*DIST
          GO TO 60
 59         AREF=AES+EMS*DIST
C
 60      AREF=DIM(AREF,0.)
         RETURN
        END
```

```
      FUNCTION ADIFF(D)
C        THE *DIFFRACTION ATTENUATION* AT DISTANCE D
C        A CONVEX COMBINATION OF SMOOTH EARTH DIFFRACTION AND
C           DOUBLE KNIFE-EDGE DIFFRACTION
C        A CALL WITH D=0 SETS UP INITIAL CONSTANTS
C
      COMMON/PROP/KWX,AREF,MDP,DIST,HG(2),WN,DH,ENS,GME,ZGND,
     X   HE(2),DL(2),THE(2)
         COMPLEX ZGND
      COMMON/PROPA/DLSA,DX,AEL,AK1,AK2,AED,EMD,AES,EMS,DLS(2),DLA,THA
C
      COMMON/SAVE/WD1,XD1,AFO,QK,AHT,XHT,SAVE(44)
C
      DATA THIRD/0.3333333/
C
       IF(D .GT. 0.) GO TO 10
C
       Q=HG(1)*HG(2)
       QK=HE(1)*HE(2)-Q
       IF(MDP .LT. 0) Q=Q+10.
       WD1=SQRT(1.+QK/Q)
       XD1=DLA+THA/GME
       Q=(1.-0.8*EXP(-DLSA/50E3))*DH
       Q=0.78*Q*EXP(-(Q/16.)**0.25)
       AFO=AMIN1(15.,2.171*ALOG(1.+4.77E-4*HG(1)*HG(2)*WN*Q))
       QK=1./CABS(ZGND)
       AHT=20.
       XHT=0.
       DO 1 J=1,2
         A=0.5*DL(J)**2/HE(J)
         WA=(A*WN)**THIRD
         PK=QK/WA
         Q=(1.607-PK)*151.0*WA*DL(J)/A
         XHT=XHT+Q
         AHT=AHT+FHT(Q,PK)
  1    CONTINUE
       ADIFF=0.
       GO TO 80
C
 10    CONTINUE
       TH=THA+D*GME
       DS=D-DLA
       Q=0.0795775*WN*DS*TH**2
       ADIFF=AKNFE(Q*DL(1)/(DS+DL(1)))+AKNFE(Q*DL(2)/(DS+DL(2)))
       A=DS/TH
       WA=(A*WN)**THIRD
       PK=QK/WA
       Q=(1.607-PK)*151.0*WA*TH+XHT
       AR=0.05751*Q-4.343*ALOG(Q)-AHT
       Q=(WD1+XD1/D)*AMIN1(((1.-0.8*EXP(-D/50E3))*DH*WN),6283.2)
       WD=25.1/(25.1+SQRT(Q))
       ADIFF=(AR-ADIFF)*WD+ADIFF+AFO
 80    RETURN
      END
```

```
      FUNCTION ALOS(D)
C        THE *LINE-OF-SIGHT ATTENUATION* AT DISTANCE D
C        A CONVEX COMBINATION OF PLANE EARTH FIELDS AND
C           DIFFRACTED FIELDS
C        A CALL WITH D=0 SETS UP INITIAL CONSTANTS
C
      COMMON/PROP/KWX,AREF,MDP,DIST,HG(2),WN,DH,ENS,GME,ZGND,
     X   HE(2),DL(2),THE(2)
         COMPLEX ZGND
      COMMON/PROPA/DLSA,DX,AEL,AK1,AK2,AED,EMD,AES,EMS,DLS(2),DLA,THA
C
      COMMON/SAVE/WLS,SAVE(49)
C
      COMPLEX R
C
      ABQ(R)=REAL(R)**2+AIMAG(R)**2
C
       IF(D .GT. 0.) GO TO 10
C
       WLS=0.021/(0.021+WN*DH/AMAX1(10E3,DLSA))
       ALOS=0.
       GO TO 80
C
   10  CONTINUE
       Q=(1.-0.8*EXP(-D/50E3))*DH
       S=0.78*Q*EXP(-(Q/16.)**0.25)
       Q=HE(1)+HE(2)
       SPS=Q/SQRT(D**2+Q**2)
       R=(SPS-ZGND)/(SPS+ZGND)*EXP(-WN*S*SPS)
       Q=ABQ(R)
       IF(Q .LT. 0.25 .OR. Q .LT. SPS) R=R*SQRT(SPS/Q)
       ALOS=EMD*D+AED
       Q=WN*HE(1)*HE(2)*2./D
       ALOS=(-4.343*ALOG(ABQ(CMPLX(COS(Q),-SIN(Q))+R))-ALOS)*WLS+ALOS
C
   80  RETURN
      END
```

```
      FUNCTION ASCAT(D)
C        THE *SCATTER ATTENUATION* AT DISTANCE D
C        USES AN APPROXIMATION TO THE METHODS OF NBS TN101 WITH
C           CHECKS FOR INADMISSABLE SITUATIONS
C        FOR PROPER OPERATION, THE LARGER DISTANCE (D=D6)
C           MUST BE THE FIRST CALLED
C        A CALL WITH D=0 SETS UP INITIAL CONSTANTS
C
      COMMON/PROP/KWX,AREF,MDP,DIST,HG(2),WN,DH,ENS,GME,ZGND,
     X   HE(2),DL(2),THE(2)
         COMPLEX ZGND
      COMMON/PROPA/DLSA,DX,AEL,AK1,AK2,AED,EMD,AES,EMS,DLS(2),DLA,THA
C
      COMMON/SAVE/AD,RR,ETQ,H0S,SAVE(46)
C
       IF(D .GT. 0.) GO TO 10
C
       AD=DL(1)-DL(2)
       RR=HE(2)/HE(1)
       IF(AD) 1,2,2
  1       AD=-AD
          RR=1./RR
  2    ETQ=(5.67E-6*ENS-2.32E-3)*ENS+0.031
       H0S=-15.
       ASCAT=0.
       GO TO 80
C
 10    CONTINUE
       IF(H0S .GT. 15.) GO TO 12
       TH=THE(1)+THE(2)+D*GME
       R2=2.*WN*TH
       R1=R2*HE(1)
       R2=R2*HE(2)
       IF(R1 .GT. 0.2 .OR. R2 .GT. 0.2) GO TO 11
          ASCAT=1001.
          GO TO 80
 11    SS=(D-AD)/(D+AD)
       Q=RR/SS
       SS=AMAX1(0.1,SS)
       Q=AMIN1(AMAX1(0.1,Q),10.)
       Z0=(D-AD)*(D+AD)*TH*0.25/D
       ET=(ETQ*EXP(-AMIN1(1.7,Z0/8.0E3)**6)+1.)*Z0/1.7556E3
       ETT=AMAX1(ET,1.)
       H0=(H0F(R1,ETT)+H0F(R2,ETT))*0.5
       H0=H0+AMIN1(H0,(1.38-ALOG(ETT))*ALOG(SS)*ALOG(Q)*0.49)
       H0=DIM(H0,0.)
       IF(ET .LT. 1.) H0=ET*H0+(1.-ET)*4.343*ALOG(((1.+1.4142/R1)*
     X    (1.+1.4142/R2))**2*(R1+R2)/(R1+R2+2.8284))
       IF(H0 .LE. 15. .OR. H0S .LT. 0.) GO TO 13
 12       H0=H0S
 13    H0S=H0
       TH=THA+D*GME
       ASCAT=AHD(TH*D)+4.343*ALOG(47.7*WN*TH**4)-
     X    0.1*(ENS-301.)*EXP(-TH*D/40E3)+H0
 80    RETURN
      END
```

```
      FUNCTION AKNFE(V2)
C        KNIFE-EDGE DIFFRACTION
C        THE FRESNEL INTEGRAL AS A FUNCTION OF V**2
C
       IF(V2 .GT. 5.76) GO TO 2
    1     AKNFE=6.02+9.11*SQRT(V2)-1.27*V2
        GO TO 8
    2     AKNFE=12.953+4.343*ALOG(V2)
    8  RETURN
      END
```

```
      FUNCTION FHT(X,PK)
C       THE HEIGHT GAIN OVER A SMOOTH SPHERICAL EARTH
C       TO BE USED IN THE *THREE RADII* METHOD
C
       IF(X .LT. 200.) GO TO 2
          FHT=0.05751*X-4.343*ALOG(X)
        IF(X .GE. 2000.) GO TO 8
          W=0.0134*X*EXP(-0.005*X)
          FHT=(1.-W)*FHT+W*(17.372*ALOG(X)-117.)
        GO TO 8
 2        IF(PK .GT. 1.E-5) GO TO 3
          IF(X .GT. 1.) GO TO 4
          FHT=-117.
        GO TO 8
 3        W=-ALOG(PK)
          IF(X*W**3 .GT. 5495.) GO TO 4
          FHT=2.5E-5*X**2/PK-8.686*W-15.
        GO TO 8
 4        FHT=17.372*ALOG(X)-117.
 8    RETURN
      END
```